```
(* R_RRT *)

(* force analysis using dyad RRT *)

Apply[Clear,Names["Global`*"]];
Off[General::spell];
Off[General::spell1];

(* Input data *)
pi = N [ Pi ] ;
n = 400; (* rpm *)
omega = n pi/30 ; (* rad/s *)

rule = {AB->0.18, BC->0.7,
h->0.01, d->0.01,
hSlider->0.02, wSlider->0.05,
ro->8000, g->9.807, fe->1000.,
phi[t]->pi/4, phi'[t]->omega, phi''[t]->0}

(* Position analysis *)

(* Position of joint A *)
xA = yA = 0 ;
rA = { xA, yA, 0} ;

(* Position, velocity and acceleration of joint B *)
xB = AB Cos[phi[t]] ;
yB = AB Sin[phi[t]] ;
rB = { xB, yB, 0} ;
vB = D[rB,t] ;
aB = D[D[rB,t],t] ;

(* Angular velocity and acceleration of the link 1 *)
omega1 = {0, 0, omega} ;
alpha1 = {0, 0, phi''[t]} ;

(* Angular velocity and acceleration of the link 2 and 3 *)

yC=0;
xC=xB+Sqrt[BC^2-(yC-yB)^2];

(* Position of joint C *)
rC = { xC, yC, 0} ;

mBC = ( yB - yC ) / ( xB - xC ) ;
phi2 = ArcTan[mBC];
omega2 = {0, 0, D[phi2,t]};
alpha2 = {0, 0, D[D[phi2,t],t]};
phi3=0;
alpha3={0,0,0};

rC1 = rB/2 ;
vC1 = vB/2 ;
aC1 = aB/2 ;

rC2 = (rB+rC)/2;
vC2 = D[rC2,t];
aC2 = D[vC2,t];

rC3 = rC;
vC3 = D[rC3,t];
aC3 = D[vC3,t] ;

pxB=xB/.rule;
```

```
pyB=yB/.rule;
pxC=xC/.rule;
pyC=yC/.rule;

Print["AC1=",rC1/.rule,"[m]; AB=",rB/.rule,"[m]"];
Print["AC2=",rC2/.rule,"[m]; AC3=AC=",rC3/.rule,"[m]"];

Print["aC1=",aC1/.rule,"[m/s^2];   alpha1=",alpha1/.rule,"[rad/s^2]"];
Print["aC2=",aC2/.rule,"[m/s^2];   alpha2=",alpha2/.rule,"[rad/s^2]"];
Print["aC3=aC",aC3/.rule,"[m/s^2];   alpha3=",alpha3/.rule,"[rad/s^2]"];

Fe=-Sign[vC3[[1]]]{fe,0,0}/.rule;
Print["Fext=",Fe,"[N]"];

m1 = ro AB h d /.rule ;
IC1 = m1 (AB^2+h^2)/12 /.rule ;
m2 = ro BC h d /.rule ;
IC2 = m2 (BC^2+h^2)/12 /.rule ;
m3 = ro hSlider wSlider d /.rule ;
IC3 = m3 (hSlider^2+wSlider^2)/12 /.rule ;

Print["m1=",m1,"[kg]; m2=",m2,"[kg]; m3=",m3,"[kg]"];
Print["IC1=",IC1,"[kg m^2]; IC2=",IC2,"[kg m^2]; IC3=",IC3,"[kg m^2]"];

Fin1 = - m1 aC1 /.rule ;
G1 = {0, -m1*g, 0} /.rule ;
F1 = ( Fin1 + G1 ) /.rule ;
M1 = Min1 = - IC1 alpha1 /.rule ;

Fin2 = - m2 aC2 /.rule ;
G2 = {0, -m2*g, 0} /.rule ;
F2 = ( Fin2 + G2 ) /.rule ;
M2 = Min2 = - IC2 alpha2 /.rule ;

Fin3 = - m3 aC3 /.rule ;
G3 = {0, -m3*g, 0} /.rule ;
F3 = ( Fin3 + G3 ) /.rule ;
M3 = Min3 = - IC3 alpha3 /.rule ;

Print["F1=Fin1+G1=-m1*aC1-m1*g=",F1,"[N]; M1=Min1=0 [N m]"];
Print["F2=Fin2+G2=-m2*aC2-m2*g=",F2,"[N]"];
Print["M2=Min2=-IC2 alpha20=",M2,"[N m]"];
Print["F3=Fin3+G3=-m3*aC -m3*g=",F3,"[N]; M3=Min3=0 [N m]"];

(*Joint reactions*)
" Dyad: B(R), C(R), C(T) "
(* @B(R)*)
F12={F12x,F12y,0};
(* @C(T)*)
F03={0,F03y,0};
Print["Sum F for 2 & 3: F12+F2+F3+F03+Fext"];
eqF23=F03+F3+F2+F12+Fe;
Print["(x): ",eqF23[[1]],"=0 (1)"];
Print["(y): ",eqF23[[2]],"=0 (2)"];
Print["Sum M for 2 wrt C:  CB x F12 + CC2 x F2 + M2 = 0 "];
rCB=(rB-rC)/.rule;
rCC2=(rC2-rC)/.rule;
eqM23=Cross[rCB,F12]+Cross[rCC2,F2]+M2;
Print["(z): ",eqM23[[3]],"=0 (3)"];
solF23=Solve[{eqF23[[1]]\[Equal]0,eqF23[[2]]\[Equal]0,
        eqM23[[3]]\[Equal]0},{F12x,F12y,F03y}];
Print["From Eqs.(1)(2)(3) => F12x, F12y, F03y"];
Print["F12 = {F12x, F12y, 0} = ",F12/.solF23[[1]]," [N]"];
Print["F03 = {0, F03y, 0} = ",F03/.solF23[[1]]," [N]"];
```

```
" Link 2 "
Print["Sum F for 2: F32+F2+F12=0 => F32=-F2-F12"];
F21={F21x,F21y,0};
F01={F01x,F01y,0};
F32=-F2-F12/.solF23[[1]];
Print["F32 = ",F32," [N]"];

" LINK 1"
"Sum F for link 1: (Fin1+G)+F21+F01=0 <=> F1+F21+F01=0"
F01=-F1+F12/.solF23[[1]];
Print["F01 = {F01x, F01y, 0} = ",F01," [N]"];
"Sum M for 1 wrt A:  AB x F21 + AC1 x F1 + M1 + Mm = 0"
Mm=Cross[(rB/.rule),(F12/.solF23[[1]])]-Cross[(rC1/.rule),F1];
Print["Mmotor = {0,0,Mm} = ",Mm," [Nm]"];
```

{AB → 0.18, BC → 0.7, h → 0.01, d → 0.01, hSlider → 0.02, wSlider → 0.05, ro → 8000, g → 9.807, fe → 1000., phi[t] → 0.785398, phi′[t] → 41.8879, phi″[t] → 0}

AC1={0.0636396, 0.0636396, 0}[m]; AB={0.127279, 0.127279, 0}[m]

AC2={0.471445, 0.0636396, 0}[m]; AC3=AC={0.815611, 0, 0}[m]

aC1={-111.662, -111.662, 0}[m/s^2]; alpha1={0, 0, 0}[rad/s^2]

aC2={-224.03, -111.662, 0}[m/s^2]; alpha2={0, 0, 313.349}[rad/s^2]

aC3=aC{-224.736, 0, 0}[m/s^2]; alpha3={0, 0, 0}[rad/s^2]

Fext={1000., 0, 0}[N]

m1=0.144[kg]; m2=0.56[kg]; m3=0.08[kg]

IC1=0.00039[kg m^2]; IC2=0.0228713[kg m^2]; IC3=0.0000193333[kg m^2]

F1=Fin1+G1=-m1*aC1-m1*g={16.0793, 14.6671, 0}[N]; M1=Min1=0 [N m]

F2=Fin2+G2=-m2*aC2-m2*g={125.457, 57.0387, 0}[N]

M2=Min2=-IC2 alpha20={0, 0, -7.16671}[N m]

F3=Fin3+G3=-m3*aC -m3*g={17.9788, -0.78456, 0}[N]; M3=Min3=0 [N m]

Dyad: B(R), C(R), C(T)

Sum F for 2 & 3: F12+F2+F3+F03+Fext

(x): 1143.44 + F12x=0 (1)

(y): 56.2541 + F03y + F12y=0 (2)

Sum M for 2 wrt C: CB x F12 + CC2 x F2 + M2 = 0

(z): -34.7815 - 0.127279 F12x - 0.688331 F12y=0 (3)

From Eqs.(1)(2)(3) => F12x, F12y, F03y

F12 = {F12x, F12y, 0} = {-1143.44, 160.902, 0} [N]

F03 = {0, F03y, 0} = {0, -217.156, 0} [N]

Link 2

Sum F for 2: F32+F2+F12=0 => F32=-F2-F12

```
F32 = {1017.98, -217.941, 0} [N]
```

```
 LINK 1
```

```
Sum F for link 1: (Fin1+G)+F21+F01=0 <=> F1+F21+F01=0
```

```
F01 = {F01x, F01y, 0} = {-1159.51, 146.235, 0} [N]
```

```
Sum M for 1 wrt A:  AB x F21 + AC1 x F1 + M1 + Mm = 0
```

```
Mmotor = {0,0,Mm} = {0., 0., 166.105} [Nm]
```